
 
University of Utah 

ME EN 6510/5510 
Introduction to Finite Elements 

Fall 2005 
 

2-D ELASTICITY 
Using ANSYS for 2-D planar elasticity problems 

 
Introduction - Planar elasticity problems are either plane stress or plane strain.  Both categories 
of problems use the same elements (defaults to plane stress).  In the Ansys program, plane42 is a 
4 node isoparametric element with 2 degrees of freedom at each node. This element is a 
“workhorse” element, and is widely used.  The ANSYS program permits internal degrees of 
freedom with incompatible modes in this element.  This feature usually leads to faster 
convergence (i.e. better accuracy), but can be suppressed if desired.  The following problem will 
illustrate how to use the program for planar problems. ANSYS also provides an advanced 
version of the plane 42 element. These are the Plane 82 elements with 8 nodes (4 at the element 
corners and 4 on the element midsides).  
 
Example - Hole in a plate - The problem to be worked is shown in Fig 1.  It is a typical stress 
concentration problem, consisting of a hole in an aluminum plate.  It will be desired to calculate 
the maximum stress around the hole, which can then be used to calculate the stress concentration 
factor.  Note that symmetry reduces the calculation to 1/4 of the actual plate.  Using symmetry, 
where appropriate, is essential.   
 

 
 

Figure 1. Hole in a flat plate problem (ANSYS model geometry). 
 
The code required to build and solve this problem is given on the following pages.  It can be 
typed into a prepared input file (*.inp) or entered directly into the command line.  When you 

finish, you might want to also play with the GUI menus on your own and see if you can do the 
problem using that method as well. 
 
Ansys input file      Comments 

/title, your title 
 et,1,42 ! element  plane42, a 4 node  isoparametric  planar  element 
 mp,ex,1,10e6 ! sets modulus  
 mp,prxy,1,.33 ! sets Poisson’s ratio 
 k,1,0.5,0 ! a “key point” 
 k,2,4.0,0 ! a “key point” 
 k,3,6,0 ! a “key point” 
 k,4,6,4 ! a “key point” 
 k,5,4,4  
 k,6,0,4 ! a “key point” (note that the point is at 0.0,4.0) 
 k,7,0,0.5 
 csys,1 ! changes from x,y (the default) to r, θ global cylindrical   
   ! coordinate system.  Use on-line help for other options. 
 k,8,0.5,45 ! places key point no. 8 at  r=0.5, θ=45 degrees 
 csys  ! changes active system to the global coordinate system ( default 
   ! option for csys command (Note: csys,0 is also correct)) 
 L,1,2,10,10 ! puts a “line” with 10 segments  between pts 1 & 2.  The last   
  ! segment  is 10 times  larger than the first. This is done to obtain  
   ! proper mesh which is sufficiently refined at regions of high 
   ! stress gradients and less refined at low gradient areas. It takes a 
   ! few trials to get the right mesh depending on the complexity of 
   ! the given problem. 
 L,8,5,10,10  !  
 L,7,6,10,10 !  
 csys,1 ! cyl. coord, so that the lines below  match  the hole 
 L,1,8,5 ! 5 segments  from 1 to 8, equally  spaced  (the default) 
 L,8,7,5 ! 5 segments  from 8 to 7, equally  spaced (the default) 
 csys ! changes active system to the global coordinate system 
 L,2,5,5 
 L,5,6,5 
 L,2,3,4 
 L,3,4,5 
 L,4,5,4 
 a,1,2,5,8 ! creates an area with key pts as corners (order is important) 
 a,8,5,6,7 ! creates an area with key pts as corners 
 a,2,3,4,5 
 amesh,all ! This creates the element mesh 
 nsel,s,loc,x,0,0 ! select  nodes on y axis (x=0) 
 dsym,symm,x ! define a symmetry  b.c. for these nodes (same as constraining  
   ! ‘ux’ displacements using ‘d’ command. 
 nsel,s,loc,y,0,0 ! select nodes  on x axis (y=0) (from all nodes) 
 dsym,symm,y ! define a symmetry  b.c. for these nodes 



 nsel,s,loc,x,6,6 ! select nodes  on right side of plate  (from all nodes) 
 sf,all,press,-1000 ! apply tension of P=1000 units to right side of plate 
 nsel,all ! reselect all nodes 
 /psf,press,2 ! show pressure as arrows on plots 
 /pbc,u,1 ! show disp. constraints  on plots 
 /rep ! replot (similar to refresh screen) 
 
 
Once the input file has been prepared as above, check to see that it is ok by typing in the 
following commands to ANSYS. 
 /prep7 !the input phase of ansys 
 /show, x11 !necessary if you are going to do plots  
 /input,filename.inp !Filename.inp  is the name  of your input file (8 letters max for   
  !filename).  All of the commands in the input file will be  
  !executed. 
 
The following commands are then used to examine your model to see that it is correct and what 
you want. 
 klist ! lists all keypoints 
 lplot ! gives a plot of the line segments 
 aplot ! plots the area.  The divisions are not elements 
 eplot ! gives a plot of elements  (a good diagnostic  tool)  
 /pbc,u,1 ! 1 shows the b.c.’s in plots, 0 turns this feature off 
 /pnum,node,1 ! 1 shows the node  no’s in plots, 0 turns this off 
 nplot ! plots the nodes 
 
 If the model  is not correct, go to the editor and fix the input file, and re-enter it as above.  If 
correct, proceed  as follows. 
 
 wsort,y ! used to minimize  bandwidth, good for larger problems.  For  

! large problems you need to try a sort in each direction, ansys  
! will  use the smallest bandwidth. 

 finish ! exits from prep7 
 
 /solu ! initiates solution phase 
 solve ! runs the solution 
 finish ! after solution, exits from solution phase 
 
 /post1 ! this enters the post processing  phase, to examine answers 
 /show,x11 ! for x windows display 
 set ! required directive 
 /edge,1,1 ! Plots model edges (optional) 
 pldisp,1 ! plots the original  and deformed  mesh, an important  diagnostic 
 plnsol,s,x ! plots a contour plot of stresses in the x direction.  Other options   
  ! sx,sy,sxy.  Good diagnostic.  
 plnsol,s,1 ! plots a contour  plot of sig1 principal stress 



 csys,1 ! change back to polar coordinates 
 nsel,s,loc,x,0,2 ! selects only nodes within a radius of 2 from the origin   
   ! (restore all nodes with “nsel,all”) 
 esln,s     ! select all elements attached to selected nodes 
 csys 
 plnsol,s,1 ! contour  plot near the region of high stresses 
 /pnum,node,1 ! 1 shows the node no’s in plots, 0 turns this off 
 nplot ! Plots the node numbers.  Note which numbers are along the   
  ! cross-section where the stress is highest 
 /edge,1,0 ! turns the edge  feature off.  Necessary for the next plot 
 lpath,72,78 ! defines a path for detailed  plotting,  based on node numbers 
 pdef,sig1,s,1 ! labels  the principal  stress along  the defined  path as sig1 
 plpath,sig1 ! plots sig1 along the path  from node 72 to node 78.  The   
  ! numbers were obtained  from the node plot 
 prpath,sig1 ! prints the sig1 values, if you want to see them 
 finish ! exits from post1 
 
When you are done, type 

/eof ! exits from Ansys 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 


